Rotations and reflections of geometric objects in the plane - with an emphasis on animating the transformation.

```
> restart: with(plots): with(plottools):with(LinearAlgebra):
Warning, the name changecoords has been redefined
Warning, the assigned name arrow now has a global binding
> p := [<1,1>, <-1,1>, <-1,-1>, <1,-1>];
  # p := [<2,0>,<3,2>,<1,4>,<0,2>,<-1,3>,<-2,0>,<-2,-2>,<0,-2>];
```

$$p := \left[\begin{bmatrix} 1 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ -1 \end{bmatrix}, \begin{bmatrix} 1 \\ -1 \end{bmatrix}\right] \tag{1}$$

```
>
>
>
```

Construct a regular n-gon on the unit circle.

```
> nGon := n -> [seq(<cos((4*i+2-n)*Pi/(2*n)),sin((4*i+2-n)*Pi/(2*n))
  >,i=0..(n-1))];
  p := nGon(6);
```

$$nGon := n \rightarrow \left[seq\left(\langle \cos\left(\frac{1}{2} \frac{(4i+2-n)\pi}{n} \right), \sin\left(\frac{1}{2} \frac{(4i+2-n)\pi}{n} \right) \rangle, i = 0..n-1 \right) \right]$$

$$p := \left[\begin{bmatrix} \frac{1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} \frac{-1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{-1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}\right] \tag{2}$$

```
> for i from 1 to (nops(p)-1) do
    l[i] := [p[i],p[i+1]];
  end do;
  l[nops(p)] := [p[nops(p)],p[1]];
  nops(l);
```

$$l_1 := \left[\begin{bmatrix} \frac{1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \end{bmatrix}\right]$$

$$l_2 := \left[\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}\right]$$

$$l_3 := \left[\begin{bmatrix} \frac{1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} \frac{-1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}\right]$$

$$l_4 := \left[\begin{bmatrix} \frac{-1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \end{bmatrix}\right]$$

$$l_5 := \left[\left[\begin{matrix} -1 \\ 0 \end{matrix} \right], \left[\begin{matrix} \frac{-1}{2} \\ -\frac{1}{2}\sqrt{3} \end{matrix} \right] \right]$$

$$l_6 := \left[\left[\begin{matrix} \frac{-1}{2} \\ -\frac{1}{2}\sqrt{3} \end{matrix} \right], \left[\begin{matrix} \frac{1}{2} \\ -\frac{1}{2}\sqrt{3} \end{matrix} \right] \right]$$

1 (3)

The colors list will allow objects with up to 16 defining points.

```
> colors := [green,blue,red,cyan,yellow,magenta,coral,violet,orange,
  sienna,navy,brown,aquamarine,gold,maroon,khaki]:
```

Create lighter shades of each color.

```
> lcolors := [COLOR(RGB,.5,1,.5),COLOR(RGB,.5,.5,1),COLOR(RGB,1,.5,
  .5),COLOR(RGB,.5,1,1),COLOR(RGB,1,1,.5),COLOR(RGB,1,.5,1),COLOR
  (RGB,1,0.7490196100,.5),COLOR(RGB,0.6549019600,0.5921568650,
  0.6549019600),COLOR(RGB,.9,0.5980392150,0.5980392150),COLOR(RGB,
  0.7784313750,0.7098039200,0.5686274500),COLOR(RGB,0.5686274500,
  0.5686274500,0.7784313750),COLOR(RGB,0.8235294100,0.5823529400,
  0.5823529400),COLOR(RGB,0.7196078450,0.9294117650,0.7882352950),
  COLOR(RGB,.9,0.7490196100,0.5980392150),COLOR(RGB,0.7784313750,
  0.5686274500,0.7098039200),COLOR(RGB,0.8117647050,0.8117647050,
  0.6862745100)]:
```

Show the colors with the lighter shades in the order they are used in plots.

```
> for i from 1 to 16 do
    tp[i] := plot([t,3,t=i..i+.5],0..17,0..10,color=colors[i],
  thickness=15):
    tpl[i] := plot([t,2.5,t=i..i+.5],0..17,0..10,color=lcolors[i],
  thickness=15):
  end do:
  display([seq(tp[i],i=1..16),seq(tpl[i],i=1..16)]);
```




```
> for i from 1 to nops(p) do
    lp[i] := pointplot(l[i],color = colors[i], thickness = 5,
  connect=true):
  end do:

> display([seq(lp[i],i=1..nops(p))],scaling=constrained,axes=none);
```



```
> rotmat := alpha -> <<cos(alpha), sin(alpha)> | <-sin(alpha), cos
  (alpha)>>;
```

$$rotmat := \alpha \rightarrow \langle\langle \cos(\alpha), \sin(\alpha)\rangle | \langle -\sin(\alpha), \cos(\alpha)\rangle\rangle \tag{4}$$

```
> rot90 := rotmat(Pi/2);
```

$$rot90 := \begin{bmatrix} 0 & -1 \\ 1 & 0 \end{bmatrix} \tag{5}$$

```
> multmatbylist :=(multmat, listofvecs)->
    map((x,y)-> y.x,listofvecs,multmat):
> rotateObj := (listoflistofvecs,ang) -> seq(multmatbylist(rotmat
  (ang),listoflistofvecs[i]),i=1..nops(p));
```

$$rotateObj := (listoflistofvecs, ang) \rightarrow seq\left(multmatbylist\left(rotmat(ang), listoflistofvecs_i\right), i = 1..nops(p)\right) \tag{6}$$

```
> ObjRot1:= rotateObj(l,Pi/3);
  l[1];
  ObjRot1[1];
  type([ObjRot1],list);
  [ObjRot1][1];
  l[1];
```

$$ObjRot1 := \left[\begin{bmatrix}1\\0\end{bmatrix}, \begin{bmatrix}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}\right], \left[\begin{bmatrix}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}, \begin{bmatrix}\frac{-1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}\right], \left[\begin{bmatrix}\frac{-1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}, \begin{bmatrix}-1\\0\end{bmatrix}\right], \left[\begin{bmatrix}-1\\0\end{bmatrix}, \begin{bmatrix}\frac{-1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix}\right], \left[\begin{bmatrix}\frac{-1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix}, \begin{bmatrix}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix}\right], \left[\begin{bmatrix}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix}, \begin{bmatrix}1\\0\end{bmatrix}\right]$$

$$\left[\begin{bmatrix}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix}, \begin{bmatrix}1\\0\end{bmatrix}\right]$$

$$\left[\begin{bmatrix}1\\0\end{bmatrix}, \begin{bmatrix}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}\right]$$

$$true$$

$$\left[\begin{bmatrix}1\\0\end{bmatrix}, \begin{bmatrix}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}\right]$$

$$\left[\begin{bmatrix}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix}, \begin{bmatrix}1\\0\end{bmatrix}\right] \qquad (7)$$

```
> for i from 1 to nops(p) do
    ObjRotp[i] := pointplot(ObjRot1[i],color = colors[i], thickness
= 5,connect=true):
end do:
display([seq(ObjRotp[i],i=1..nops(p))],scaling=constrained);
```




```
> display([seq(display([seq(pointplot(l[i],color = lcolors[i],
thickness = 5,connect=true),i=1..nops(p)),seq(pointplot
(multmatbylist(rotmat((counta/18)*Pi/2),l[i]),color=colors[i],
connect=true,thickness=5),i=1..nops(p))]), counta=0..18)],
insequence = true);
```




```
> print(l);
  nops(l); l[1];l[2];
```

$$table\left(\left[1=\left[\left[\begin{array}{c}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{array}\right],\left[\begin{array}{c}1\\ 0\end{array}\right]\right],2=\left[\left[\begin{array}{c}1\\ 0\end{array}\right],\left[\begin{array}{c}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{array}\right]\right],3=\left[\left[\begin{array}{c}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{array}\right],\left[\begin{array}{c}\frac{-1}{2}\\ \frac{1}{2}\sqrt{3}\end{array}\right]\right],\right.\right.$$

$$\left.5=\left[\left[\begin{array}{c}-1\\ 0\end{array}\right],\left[\begin{array}{c}\frac{-1}{2}\\ -\frac{1}{2}\sqrt{3}\end{array}\right]\right],4=\left[\left[\begin{array}{c}\frac{-1}{2}\\ \frac{1}{2}\sqrt{3}\end{array}\right],\left[\begin{array}{c}-1\\ 0\end{array}\right]\right],\right.$$

$$\left.\left.6=\left[\left[\begin{array}{c}\frac{-1}{2}\\ -\frac{1}{2}\sqrt{3}\end{array}\right],\left[\begin{array}{c}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{array}\right]\right]\right]\right)$$

$$\left[\left[\begin{bmatrix}\frac{1}{2}\\ -\frac{1}{2}\sqrt{3}\end{bmatrix},\begin{bmatrix}1\\0\end{bmatrix}\right],\left[\begin{bmatrix}1\\0\end{bmatrix},\begin{bmatrix}\frac{1}{2}\\ \frac{1}{2}\sqrt{3}\end{bmatrix}\right]\right] \tag{8}$$

```
> AnimRotate := proc(l,ang)
  local NumSteps;
  NumSteps := ceil(36*abs(ang)/Pi):
  display([seq(display([seq(pointplot(l[i],color = lcolors[i],
  thickness = 5,connect=true),i=1..nops(p)),seq(pointplot
  (multmatbylist(rotmat((counta/NumSteps)*ang),l[i]),color=colors[i]
  ,connect=true,thickness=5),i=1..nops(p))]), counta=0..NumSteps)],
  scaling=constrained, insequence = true)
  end proc;
```

AnimRotate := **proc**(*l*, *ang*) (9)

local *NumSteps*;

NumSteps := ceil((36 * abs(*ang*)) / Pi);

display([*seq*(*display*([*seq*(*pointplot*(

l[*i*], *color* = *lcolors*[*i*], *thickness* = 5, *connect* = *true*),

i = 1 ..*nops*(*p*)), *seq*(*pointplot*(*multmatbylist*(*rotmat*((*counta* * *ang*) / *NumSteps*),

l[*i*]), *color* = *colors*[*i*], *connect* = *true*, *thickness* = 5), *i* = 1 ..*nops*(*p*))]),

counta = 0 ..*NumSteps*)], *scaling* = *constrained*, *insequence* = *true*)

end proc

```
> AnimRotate(l,Pi/3);
```



Now work on the reflections. Basic reflection matrix reflects in steps across x-axis, but we represent this as rotation around the x-axis in R3.

```
> refinxmat := alpha -> <<1,0,0>|<0,cos(alpha),sin(alpha)>|<0,-sin
  (alpha),cos(alpha)>>;
```

$$refinxmat := \alpha \rightarrow \langle\langle 1, 0, 0\rangle|\langle 0, \cos(\alpha), \sin(\alpha)\rangle|\langle 0, -\sin(\alpha), \cos(\alpha)\rangle\rangle \tag{10}$$

```
> refinxmat(Pi);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{bmatrix} \tag{11}$$

To reflect in the line at angle theta we need to conjugate by the change of basis matrix

```
> chbasis := theta -> <<cos(theta),sin(theta),0>|<-sin(theta),cos
  (theta),0>|<0,0,1>>;
  refmat := (theta,alpha) -> chbasis(theta).refinxmat(alpha).
  (chbasis(theta))^(-1);
```

$$chbasis := \theta \rightarrow \langle\langle \cos(\theta), \sin(\theta), 0\rangle|\langle -\sin(\theta), \cos(\theta), 0\rangle|\langle 0, 0, 1\rangle\rangle$$

$$refmat := (\theta, \alpha) \rightarrow chbasis(\theta).refinxmat(\alpha).\left(\frac{1}{chbasis(\theta)}\right) \tag{12}$$

```
> refmat(3*Pi/4,Pi);
```

$$\begin{bmatrix} 0 & -1 & 0 \\ -1 & 0 & 0 \\ 0 & 0 & -1 \end{bmatrix} \tag{13}$$

To reflect a vector we embed into R3, apply the reflection matrix, and then project back to R2.

```
> projembmultmatbylist :=(multmat, listofvecs)->
    map((x,y)-> <(y.<x[1],x[2],0>)[1],(y.<x[1],x[2],0>)[2]>,
  listofvecs,multmat):

> projembmultmatbylist(refmat(Pi/4,Pi),l[1]);
```

$$\left[\begin{bmatrix} -\frac{1}{2}\sqrt{3} \\ \frac{1}{2} \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix}\right] \tag{14}$$

```
> reflectObj := (listoflistofvecs,ang) -> seq(projembmultmatbylist
  (refmat(ang,Pi),listoflistofvecs[i]),i=1..nops(p));
```

$$reflectObj := (listoflistofvecs, ang) \rightarrow seq\left(projembmultmatbylist\left(refmat(ang, \pi), listoflistofvecs_i\right), i = 1..nops(p)\right) \tag{15}$$

```
> refObj1 := reflectObj(l,Pi/6);
```

$$refObj1 := \left[\left[\begin{bmatrix} \frac{-1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} \frac{1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}\right], \left[\begin{bmatrix} \frac{1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \end{bmatrix}\right], \left[\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}\right], \left[\begin{bmatrix} \frac{1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} \frac{-1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}\right], \left[\begin{bmatrix} \frac{-1}{2} \\ -\frac{1}{2}\sqrt{3} \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \end{bmatrix}\right], \left[\begin{bmatrix} -1 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{-1}{2} \\ \frac{1}{2}\sqrt{3} \end{bmatrix}\right]\right] \tag{16}$$

```
> for i from 1 to nops(p) do
    refObj1p[i] := pointplot(refObj1[i],color = colors[i],
  thickness = 5,connect=true):
  end do:
  display([seq(refObj1p[i],i=1..nops(p))],scaling=constrained);
```




```
>
> AnimReflect := proc(l,ang)
  display([seq(display([plot([cos(ang)*t,sin(ang)*t,t=-1.1..1.1],
  linestyle=DOT,thickness=2,color=black),seq(pointplot(l[i],color =
  lcolors[i], thickness = 5,connect=true),i=1..nops(p)),seq
  (pointplot(projembmultmatbylist(refmat(ang,(counta/18)*Pi),l[i]),
  color=colors[i],connect=true,thickness=5),i=1..nops(p))]), counta=
  0..18)], insequence = true,scaling=constrained)
  end proc;
```

$$AnimReflect := \mathbf{proc}(l, ang) \tag{17}$$

$$display([seq(display([plot([\cos(ang)*t, \sin(ang)*t, t=-1.1..1.1], linestyle=DOT, thickness=2, color=black), seq(pointplot(l[i], color=lcolors[i], thickness=5, connect=true), i=1..nops(p))), seq(pointplot(projembmultmatbylist(refmat(ang, 1/18*counta*Pi), l[i]), color=colors[i], connect=true, thickness=5), i=1..nops(p))]), counta=0..18)], insequence=true, scaling=constrained)$$

end proc

```
> AnimReflect(l,Pi/3);
```




```
> AnimReflect(l,Pi/6);
```




```
> l;
```

$$l \tag{18}$$

```
> ObjRot1;
```

$$\left[\left[\begin{bmatrix}1\\0\end{bmatrix},\begin{bmatrix}\frac{1}{2}\\\frac{1}{2}\sqrt{3}\end{bmatrix}\right],\left[\left[\begin{bmatrix}\frac{1}{2}\\\frac{1}{2}\sqrt{3}\end{bmatrix},\begin{bmatrix}\frac{-1}{2}\\\frac{1}{2}\sqrt{3}\end{bmatrix}\right],\left[\left[\begin{bmatrix}\frac{-1}{2}\\\frac{1}{2}\sqrt{3}\end{bmatrix},\begin{bmatrix}-1\\0\end{bmatrix}\right],\left[\left[\begin{bmatrix}-1\\0\end{bmatrix},\begin{bmatrix}\frac{-1}{2}\\-\frac{1}{2}\sqrt{3}\end{bmatrix}\right],\left[\left[\begin{bmatrix}\frac{-1}{2}\\-\frac{1}{2}\sqrt{3}\end{bmatrix},\begin{bmatrix}\frac{1}{2}\\-\frac{1}{2}\sqrt{3}\end{bmatrix}\right],\left[\left[\begin{bmatrix}\frac{1}{2}\\-\frac{1}{2}\sqrt{3}\end{bmatrix},\begin{bmatrix}1\\0\end{bmatrix}\right]\right]\right]\right]\right] \tag{19}$$

```
>
```